Rules for integrands of the form $(d\,x)^m\,(a+b\,\mathrm{Log}[c\,x^n])^p$

1: $\int \frac{(a+b\,\mathrm{Log}[c\,x^n])^p}{x}\,dx$

Reference: CRC 491

Derivation: Integration by substitution

Basis: $\frac{F[a+b\,\mathrm{Log}[c\,x^n]]}{x} == \frac{1}{b\,n}\,\mathrm{Subst}[F[x],\,x,\,a+b\,\mathrm{Log}[c\,x^n]]\,\partial_x\,(a+b\,\mathrm{Log}[c\,x^n])$

Rule:

$$\int \frac{(a+b\,\mathrm{Log}[c\,x^n])^p}{x}\,dx \longrightarrow \frac{1}{b\,n}\,\mathrm{Subst}\left[\int x^p\,dx,\,x,\,a+b\,\mathrm{Log}[c\,x^n]\right]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*x_^n_.])/x_,x_Symbol] :=
  (a+b*Log[c*x^n])^2/(2*b*n) /;
FreeQ[{a,b,c,n},x]
```

```
Int[(a_.+b_.*Log[c_.*x_^n_.])^p_./x_,x_Symbol] :=
  1/(b*n)*Subst[Int[x^p,x],x,a+b*Log[c*x^n]] /;
FreeQ[{a,b,c,n,p},x]
```

2. $\int (d\,x)^m \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p \, dx$ when $m \neq -1 \wedge p>0$

1: $\int (d\,x)^m \left(a+b\,\text{Log}\left[c\,x^n\right]\right) dx$ when $m \neq -1 \wedge a\,(m+1)-b\,n=0$

- Note: Optional rule for special case returns a single term.

- Rule: If $m \neq -1$, then

$$\int (d\,x)^m \left(a+b\,\text{Log}\left[c\,x^n\right]\right) dx \longrightarrow \frac{b\,(d\,x)^{m+1}\,\text{Log}\left[c\,x^n\right]}{d\,(m+1)}$$

- Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  b*(d*x)^(m+1)*Log[c*x^n]/(d*(m+1)) /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[m,-1] && EqQ[a*(m+1)-b*n,0]
```

2: $\int (d\,x)^m \left(a+b\,\text{Log}[c\,x^n]\right)^p dx$ when $m \neq -1 \land p > 0$

Reference: G&R 2.721.1, CRC 496, A&S 4.1.51

Derivation: Integration by parts

Basis: $\partial_x \left(a+b\,\text{Log}[c\,x^n]\right)^p == \frac{b\,n\,p\,\left(a+b\,\text{Log}[c\,x^n]\right)^{p-1}}{x}$

Rule: If $m \neq -1 \land p > 0$, then

$$\int (d\,x)^m \left(a+b\,\text{Log}[c\,x^n]\right)^p dx \longrightarrow \frac{(d\,x)^{m+1}\left(a+b\,\text{Log}[c\,x^n]\right)^p}{d\,(m+1)} - \frac{b\,n\,p}{m+1}\int (d\,x)^m \left(a+b\,\text{Log}[c\,x^n]\right)^{p-1} dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  (d*x)^(m+1)*(a+b*Log[c*x^n])/(d*(m+1)) - b*n*(d*x)^(m+1)/(d*(m+1)^2) /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[m,-1]
```

```
Int[(d_.*x_)^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  (d*x)^(m+1)*(a+b*Log[c*x^n])^p/(d*(m+1)) - b*n*p/(m+1)*Int[(d*x)^m*(a+b*Log[c*x^n])^(p-1),x] /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[m,-1] && GtQ[p,0]
```

3: $\int (d\,x)^m \left(a+b\,\text{Log}[c\,x^n]\right)^p dx$ when $m \neq -1 \land p < -1$

Reference: G&R 2.724.1, CRC 495

Derivation: Inverted integration by parts

Rule: If $m \neq -1 \land p < -1$, then

$$\int (d\,x)^m \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^p dx \longrightarrow \frac{(d\,x)^{m+1}\left(a+b\,\text{Log}\left[c\,x^n\right]\right)^{p+1}}{b\,d\,n\,(p+1)} - \frac{m+1}{b\,n\,(p+1)}\int (d\,x)^m \left(a+b\,\text{Log}\left[c\,x^n\right]\right)^{p+1} dx$$

– Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*Log[c*x^n])^(p+1)/(b*d*n*(p+1)) - (m+1)/(b*n*(p+1))*Int[(d*x)^m*(a+b*Log[c*x^n])^(p+1),x] /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[m,-1] && LtQ[p,-1]
```

4. $\int \frac{(d\,x)^m}{\text{Log}\left[c\,x^n\right]} dx$ when $m == n-1$

1: $\int \frac{x^m}{\text{Log}\left[c\,x^n\right]} dx$ when $m == n-1$

Derivation: Integration by substitution

– Note: The resulting antiderivative of this unessential rule is expressed in terms of `LogIntegral` instead of `ExpIntegralEi`.

– Rule: If $m == n-1$, then

$$\int \frac{x^m}{\text{Log}\left[c\,x^n\right]} dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\int \frac{1}{\text{Log}[c\,x]} dx,\,x,\,x^n\right]$$

– Program code:

```
Int[x_^m_./Log[c_.*x_^n_],x_Symbol] :=
  1/n*Subst[Int[1/Log[c*x],x],x,x^n] /;
FreeQ[{c,m,n},x] && EqQ[m,n-1]
```

2: $\int \frac{(d\,x)^m}{\text{Log}[c\,x^n]}\,dx$ when $m == n - 1$

Derivation: Piecewise constant extraction

Rule: If $m == n - 1$, then

$$\int \frac{(d\,x)^m}{\text{Log}[c\,x^n]}\,dx \rightarrow \frac{(d\,x)^m}{x^m}\int \frac{x^m}{\text{Log}[c\,x^n]}\,dx$$

Program code:

```
Int[(d_*x_)^m_./Log[c_.*x_^n_],x_Symbol] :=
  (d*x)^m/x^m*Int[x^m/Log[c*x^n],x] /;
FreeQ[{c,d,m,n},x] && EqQ[m,n-1]
```

5: $\int x^m\,(a + b\,\text{Log}[c\,x])^p\,dx$ when $m \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $m \in \mathbb{Z}$, then $x^m\,F[\text{Log}[c\,x]] == \frac{1}{c^{m+1}}\,\text{Subst}\left[e^{(m+1)\,x}\,F[x],\,x,\,\text{Log}[c\,x]\right]\,\partial_x\,\text{Log}[c\,x]$

Rule: If $m \in \mathbb{Z}$, then

$$\int x^m\,(a + b\,\text{Log}[c\,x])^p\,dx \rightarrow \frac{1}{c^{m+1}}\,\text{Subst}\left[\int e^{(m+1)\,x}\,(a + b\,x)^p\,dx,\,x,\,\text{Log}[c\,x]\right]$$

Program code:

```
Int[x_^m_.*(a_.+b_.*Log[c_.*x_])^p_,x_Symbol] :=
  1/c^(m+1)*Subst[Int[E^((m+1)*x)*(a+b*x)^p,x],x,Log[c*x]] /;
FreeQ[{a,b,c,p},x] && IntegerQ[m]
```

6: $\int (d\,x)^m\,(a+b\,\text{Log}[c\,x^n])^p\,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{(d\,x)^{m+1}}{(c\,x^n)^{\frac{m+1}{n}}} = 0$

Basis: $\frac{(c\,x^n)^k\,F[\text{Log}[c\,x^n]]}{x} = \frac{1}{n}\,\text{Subst}\left[e^{k\,x}\,F[x],\,x,\,\text{Log}[c\,x^n]\right]\,\partial_x \text{Log}[c\,x^n]$

Rule:

$$\int (d\,x)^m\,(a+b\,\text{Log}[c\,x^n])^p\,dx \longrightarrow \frac{(d\,x)^{m+1}}{d\,(c\,x^n)^{\frac{m+1}{n}}}\int \frac{(c\,x^n)^{\frac{m+1}{n}}\,(a+b\,\text{Log}[c\,x^n])^p}{x}\,dx \longrightarrow \frac{(d\,x)^{m+1}}{d\,n\,(c\,x^n)^{\frac{m+1}{n}}}\,\text{Subst}\left[\int e^{\frac{m+1}{n}x}\,(a+b\,x)^p\,dx,\,x,\,\text{Log}[c\,x^n]\right]$$

Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_,x_Symbol] :=
  (d*x)^(m+1)/(d*n*(c*x^n)^((m+1)/n))*Subst[Int[E^((m+1)/n*x)*(a+b*x)^p,x],x,Log[c*x^n]] /;
FreeQ[{a,b,c,d,m,n,p},x]
```

P: $\int (d\,x^q)^m\,(a+b\,\text{Log}[c\,x^n])^p\,dx$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(d\,x^q)^m}{x^{m\,q}} == 0$

Rule:

$$\int (d\,x^q)^m\,(a+b\,\text{Log}[c\,x^n])^p\,dx \to \frac{(d\,x^q)^m}{x^{m\,q}}\int x^{m\,q}\,(a+b\,\text{Log}[c\,x^n])^p\,dx$$

Program code:

```
Int[(d_.*x_^q_)^m_*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  (d*x^q)^m/x^(m*q)*Int[x^(m*q)*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d,m,n,p,q},x]
```

```
Int[(d1_.*x_^q1_)^m1_*(d2_.*x_^q2_)^m2_*(a_.+b_.*Log[c_.*x_^n_.])^p_.,x_Symbol] :=
  (d1*x^q1)^m1*(d2*x^q2)^m2/x^(m1*q1+m2*q2)*Int[x^(m1*q1+m2*q2)*(a+b*Log[c*x^n])^p,x] /;
FreeQ[{a,b,c,d1,d2,m1,m2,n,p,q1,q2},x]
```